

Analog to Digital Conversion

INTRODUCTION

This application note describes a method for implementing analog to digital conversion on the PIC16C5X series of microcontrollers. The converter requires only five external components and is software and hardware configurable for conversion resolutions from 6 bits up to 10 bits and conversion times of 250 μ s or longer. The method is useable for both voltage and current conversion and uses a software calibration technique that compensates for time and temperature drift as well as component errors. The PIC16C5X microcontrollers are ideal for simple analog applications because:

- * Very low cost.
- * Few external components required.
- * Fully programmable. PIC16C5X microcontrollers are offered as One-Time-Programmable (OTP) EPROM devices.
- * Available off the shelf from distributors.
- * Calibration in software for improved measurement accuracy.
- * Power savings using PIC16C5X's Sleep mode.
- * PIC16C5X's output pins have large, current source/sink capability to drive LED's directly.

THEORY OF OPERATION

The application uses a capacitive charging circuit (see Figure 1) to convert the input voltage to time, which can be easily measured using a microcontroller. First, the reference voltage is applied to the input voltage to current converter (U1). The equivalent circuit is shown in Figure 2. This circuit provides a linearly variable current as a function of input voltage. The logarithmic characteristic that would occur if the input voltage was applied directly to an RC is not present. The capacitor C is charged up until the threshold on the chip input trips. This generates a software calibration value that is used to calibrate out most circuit errors, including inaccuracies in the resistor and capacitor, changes in the input threshold voltage and temperature variations. After the software calibration value is measured, the capacitor is discharged (see Figure 3) and the input voltage is connected to V_{IN}. The time to trip the threshold is measured for the input voltage and compared to the calibration value to determine the actual input voltage.

FIGURE 1 - VOLTMETER A TO D CONVERTER




FIGURE 2 - VOLTMETER MEASUREMENT CYCLE





FIGURE 3 - VOLTMETER DISCHARGE CYCLE



Analog to Digital Conversion

CIRCUIT CONFIGURATION

The values of R and C are selected based upon the number of bits of resolution required.

$$RC = (V_i \cdot T) / V_t$$

Where:

V_i = Lowest voltage to be measured (at least ten I_{sb} 's)

T = Time to do the number of bits of resolution desired

V_t = Threshold voltage of the PIC16C5X input being used

Actual value for RC should be slightly smaller than calculated to ensure that the PIC16C5X does not overcount during the measurement.

For example use a 3 volt input and 8 bits resolution with a 8 MHz clock and 6 instruction cycles per count:

$$V_i = 100 \text{ mV}$$

$$T = 256 \cdot 1/8 \text{ MHz} \cdot 4 \text{ clocks/cycle} \cdot 6 \text{ cycles} = 768 \mu\text{S}$$

$$V_t = 3.0\text{V (est)}$$

For input voltages greater than 3 volts a resistor divider network should be used to keep the maximum voltage on V_{IN} to less than 3 Volts. For best performance the reference voltage should be between 2 and 3 volts.

The circuit can also be used as a current mode A to D converter. In this case the input voltage to current converter is not needed and the reference current and input current are both routed via analog switches directly into the capacitor.

CIRCUIT PERFORMANCE


The calibration cycle removes all first order errors (offset, gain, R and C inaccuracy, power supply voltage and temperature) except the reference voltage drift. Any change in the reference voltage, including noise, between the calibration cycle and the measurement cycle may result in measurement errors. Other error sources are analog switch leakage, resistor and capacitor non-linearities, input threshold uncertainty and time measurement uncertainty (+/- one instruction cycle time). Measured performance shows the converter to be accurate within +/- 1% of full scale.

Example

Assembly code implementing the circuit of Figure 1 is listed in Appendix A: This code measures the time up to 16 bits and calculates the results using 16-bit multiply and divide subroutines. In actual applications, if measurement accuracy permits, it may be advantageous to use 8 bits. The math code can be substantially reduced and the measure time is reduced by the simpler code and shorter count.

*Author: Doug Cox
Logic Products Division*

FIGURE 4 - TRANSMISSION FLOW CHART



Analog to Digital Conversion

APPENDIX A:

MPASM B0.54

PAGE 1

VOLTMETER/AD CONVERTER PROGRAM REV 3-29-90

```
TITLE 'VOLTMETER/AD CONVERTER PROGRAM REV 3-29-90'
LIST P=16C54,F=inhx16,n=0

0008          ACCA    EQU    8
000A          ACCB    EQU    0A
000C          ACCC    EQU    0C
000E          ACCD    EQU    0E
0010          ACCE    EQU    10
0012          TMEAS   EQU    12
0014          TEMP    EQU    14

0060          VCALMS  EQU    60          ;VCAL MSB VALUE IN HEX
00A4          VCALLS  EQU    0A4        ;VCAL LSB VALUE IN HEX

          ORG 1FF
01FF 0A58          GOTO  VOLTS          ;PROGRAM CODE
          ORG      0                    ;SUBROUTINES

0000 0209          MADD    MOVF    ACCA+1,W
0001 01EB          ADDWF   ACCB+1      ;ADD LSB
0002 0603          BTFSC   3,0        ;ADD IN CARRY
0003 02AA          INCF   ACCB
0004 0208          MOVF   ACCA,W
0005 01EA          ADDWF   ACCB        ;ADD MSB
0006 0800          RETLW  0
0007 0000          NOP

0008 0915          MPY    CALL    SETUP          ;RESULTS IN B(16 MSB'S) AND C(16 LSB'S)
0009 032E          MLOOP  RRF    ACCD          ;ROTATE D RIGHT
000A 032F          RRF    ACCD+1
000B 0603          SKPNC                    ;NEED TO ADD?
000C 0900          CALL    MADD
000D 032A          RRF    ACCB
000E 032B          RRF    ACCB+1
000F 032C          RRF    ACCC
0010 032D          RRF    ACCC+1
0011 02F4          DECFSZ TEMP          ;LOOP UNTIL ALL BITS CHECKED
0012 0A09          GOTO   MLOOP
0013 0800          RETLW  0

0014 0000          NOP
0015 0C10          SETUP  MOVLW  10
0016 0034          MOVWF  TEMP
0017 020A          MOVF   ACCB,W          ;MOVE B TO D
0018 002E          MOVWF  ACCD
0019 020B          MOVF   ACCB+1,W
001A 002F          MOVWF  ACCD+1
001B 020C          MOVF   ACCC,W
001C 0030          MOVWF  ACCE
001D 020D          MOVF   ACCC+1,W
001E 0031          MOVWF  ACCE+1
001F 006A          CLRF  ACCB
0020 006B          CLRF  ACCB+1
0021 0800          RETLW  0

0022 0000          NOP
0023 0915          DIV    CALL    SETUP
0024 0C20          MOVLW  20
0025 0034          MOVWF  TEMP
0026 006C          CLRF  ACCC
0027 006D          CLRF  ACCC+1
```

2

Analog to Digital Conversion

```

0028 0403          DLOOP    CLRC
0029 0371          RLF      ACCE+1
002A 0370          RLF      ACCE
002B 036F          RLF      ACCD+1
002C 036E          RLF      ACCD
002D 036D          RLF      ACCC+1
002E 036C          RLF      ACCC
002F 0208          MOVF     ACCA,W
0030 008C          SUBWF   ACCC,W          ;CHECK IF A>C
0031 0743          SKPZ
0032 0A35          GOTO    NOCHK
0033 0209          MOVF     ACCA+1,W
0034 008D          SUBWF   ACCC+1,W      ;IF MSB EQUAL THEN CHECK LSB
0035 0703          SKPC          ;CARRY SET IF C>A
0036 0A3E          GOTO    NOGO
0037 0209          MOVF     ACCA+1,W      ;C-A INTO C
0038 00AD          SUBWF   ACCC+1
0039 0703          BTFSS  3,0
003A 00EC          DECF   ACCC
003B 0208          MOVF     ACCA,W
003C 00AC          SUBWF   ACCC
003D 0503          SETC          ;SHIFT A 1 INTO B (RESULT)
003E 036B          RLF      ACCB+1
003F 036A          RLF      ACCB
0040 02F4          DECFSZ  TEMP          ;LOOP UNTILL ALL BITS CHECKED
0041 0A28          GOTO    DLOOP
0042 0800          RETLW   0

0043 0C0E          DSCHRG  MOVLW  B'00001110'  ;DISCHARGE C (RA0 ON)
0044 0005          TRIS   5
0045 0CF7          MOVLW  OFF
0046 0034          MOVWF  TEMP
0047 02F4          LOOP   DECFSZ  TEMP          ;WAIT
0048 0A47          GOTO   LOOP
0049 0C0F          MOVLW  B'00001111'  ;ALL RA HIGH Z
004A 0005          TRIS   5
004B 0800          RETLW  0

004C 0061          M_TIME CLRF   1          ;CLEAR RTCC REGISTER
004D 0069          CLRF   ACCA+1      ;CLEAR 16 BIT COUNTER
004E 0068          CLRF   ACCA
004F 03E9          TLOOP  INCFSZ  ACCA+1
0050 0A54          GOTO   ENDCHK
0051 03E8          INCFSZ  ACCA
0052 0A54          GOTO   ENDCHK
0053 0A56          GOTO   END_M
0054 0701          ENDCHK BTFSS  1,0          ;CHECK FOR RTCC TRIP
0055 0A4F          GOTO   TLOOP
0056 0201          END_M  MOVF   1,W
0057 0800          RETLW  0

0058 0C06          VOLTS  MOVLW  B'00000110'  ;SET S2 AND S3 HIGH(ON WHEN ACTIVATED)
0059 0026          MOVWF  6
005A 0CF0          MOVLW  B'11110000'  ;ACTIVATE SWITCHES S1-S4
005B 0006          TRIS   6
005C 0C28          MOVLW  B'00101000'  ;SELECT POSITIVE EDGE FOR RTCC
005D 0002          OPTION
005E 0C00          MOVLW  B'00000000'
005F 0025          MOVWF  5          ;SET RA0 LOW (ON WHEN ACTIVATED)

0060 0943          MEAS  CALL   DSCHRG          ;CHARGE CAPACITOR TO VIN
0061 0C0A          MOVLW  B'00001010'  ;S2 AND S4 ON
0062 0026          MOVWF  6
0063 094C          CALL   M_TIME          ;MEASURE TIME
0064 0209          MOVF   ACCA+1,W
0065 0033          MOVWF  TMEAS+1      ;STORE LSB
0066 0208          MOVF   ACCA,W
0067 0032          MOVWF  TMEAS          ;STORE MSB

```

Analog to Digital Conversion

```
0068 0C05          CAL      MOVLW  B'00000101'      ;S1 AND S3 ON
0069 0026          MOVWF   6
006A 0943          CALL    DSCHRG          ;CHARGE CAPACITOR TO VREF
006B 0C09          MOVLW  B'00001001'      ;S1 AND S4 ON
006C 0026          MOVWF   6
006D 094C          CALL    M_TIME          ;MEASURE TIME

006E 0CA4          MOVLW  VCALLS
006F 002B          MOVWF   ACCB+1
0070 0C60          MOVLW  VCALMS
0071 002A          MOVWF   ACCB

0072 0908          CALL    MPY          ;MULTIPLY ACCA(TCAL) * ACCB(VREF)
0073 0213          MOVF   TMEAS+1,W
0074 0029          MOVWF  ACCA+1
0075 0212          MOVF   TMEAS,W
0076 0028          MOVWF  ACCA

0077 0923          CALL    DIV          ;DIVIDE ACCB(TCAL * V) BY ACCA(TMEAS)

0078 0A58          GOTO   VOLTS

                                END

Errors   :    0
Warnings :    0
```

2

Analog to Digital Conversion

NOTES:

WORLDWIDE SALES & SERVICE

AMERICAS

Corporate Office

Microchip Technology Inc.
2355 West Chandler Blvd.
Chandler, AZ 85224-6199
Tel: 602 786-7200 Fax: 602 786-7277
Technical Support: 602 786-7627
Web: <http://www.mchip.com/microhip>

Atlanta

Microchip Technology Inc.
500 Sugar Mill Road, Suite 200B
Atlanta, GA 30350
Tel: 770 640-0034 Fax: 770 640-0307

Boston

Microchip Technology Inc.
5 Mount Royal Avenue
Marlborough, MA 01752
Tel: 508 480-9990 Fax: 508 480-8575

Chicago

Microchip Technology Inc.
333 Pierce Road, Suite 180
Itasca, IL 60143
Tel: 708 285-0071 Fax: 708 285-0075

Dallas

Microchip Technology Inc.
14651 Dallas Parkway, Suite 816
Dallas, TX 75240-8809
Tel: 214 991-7177 Fax: 214 991-8588

Dayton

Microchip Technology Inc.
35 Rockridge Road
Englewood, OH 45322
Tel: 513 832-2543 Fax: 513 832-2841

Los Angeles

Microchip Technology Inc.
18201 Von Karman, Suite 455
Irvine, CA 92715
Tel: 714 263-1888 Fax: 714 263-1338

New York

Microchip Technology Inc.
150 Motor Parkway, Suite 416
Hauppauge, NY 11788
Tel: 516 273-5305 Fax: 516 273-5335

AMERICAS (continued)

San Jose

Microchip Technology Inc.
2107 North First Street, Suite 590
San Jose, CA 95131
Tel: 408 436-7950 Fax: 408 436-7955

ASIA/PACIFIC

Hong Kong

Microchip Technology
Unit No. 3002-3004, Tower 1
Metroplaza
223 Hing Fong Road
Kwai Fong, N.T. Hong Kong
Tel: 852 2 401 1200 Fax: 852 2 401 3431

Korea

Microchip Technology
168-1, Youngbo Bldg. 3 Floor
Samsung-Dong, Kangnam-Ku,
Seoul, Korea
Tel: 82 2 554 7200 Fax: 82 2 558 5934

Singapore

Microchip Technology
200 Middle Road
#10-03 Prime Centre
Singapore 188980
Tel: 65 334 8870 Fax: 65 334 8850

Taiwan

Microchip Technology
10F-1C 207
Tung Hua North Road
Taipei, Taiwan, ROC
Tel: 886 2 717 7175 Fax: 886 2 545 0139

EUROPE

United Kingdom

Arizona Microchip Technology Ltd.
Unit 6, The Courtyard
Meadow Bank, Furlong Road
Bourne End, Buckinghamshire SL8 5AJ
Tel: 44 0 1628 851077 Fax: 44 0 1628 850259

France

Arizona Microchip Technology SARL
2 Rue du Buisson aux Fraises
91300 Massy - France
Tel: 33 1 69 53 63 20 Fax: 33 1 69 30 90 79

Germany

Arizona Microchip Technology GmbH
Gustav-Heinemann-Ring 125
D-81739 Muenchen, Germany
Tel: 49 89 627 144 0 Fax: 49 89 627 144 44

Italy

Arizona Microchip Technology SRL
Centro Direzionale Colleoni
Palazzo Pegaso Ingresso No. 2
Via Paracelso 23, 20041
Agrate Brianza (MI) Italy
Tel: 39 039 689 9939 Fax: 39 039 689 9883

JAPAN

Microchip Technology Intl. Inc.
Benex S-1 6F
3-18-20, Shin Yokohama
Kohoku-Ku, Yokohama
Kanagawa 222 Japan
Tel: 81 45 471 6166 Fax: 81 45 471 6122

9/22/95

All rights reserved. © 1995, Microchip Technology Incorporated, USA.

Information contained in this publication regarding device applications and the like is intended through suggestion only and may be superseded by updates. No representation or warranty is given and no liability is assumed by Microchip Technology Incorporated with respect to the accuracy or use of such information, or infringement of patents or other intellectual property rights arising from such use or otherwise. Use of Microchip's products as critical components in life support systems is not authorized except with express written approval by Microchip. No licenses are conveyed, implicitly or otherwise, under any intellectual property rights. The Microchip logo and name are registered trademarks of Microchip Technology Inc. All rights reserved. All other trademarks mentioned herein are the property of their respective companies.